FOURIER Series

Univ.-Prof. Dr.-Ing. habil. Josef BETTEN
RWTH Aachen University
Templergraben 55
D-52056 A a c h e n , Germany
betten@mmw.rwth-aachen.de

Abstract

This worksheet is concerned with *FOURIER* series. Some examples are discussed using MAPLE V, Release 10.

Keywords: *FOURIER* expansion; odd and even functions; *HEAVISIDE* function; continuous functions with cusps; L-two norm

FOURIER Expansion

```
restart:
FOURIER_series:=
a[0]/2+sum(a[k]*cos(k*x)+b[k]*sin(k*x),k=1..infinity);
```

$$FOURIER\_series := \frac{1}{2} a_0 + \left(\sum_{k=1}^{\infty} \left(a_k \cos(k\,x) + b_k \sin(k\,x) \right) \right)$$

```
> a[k]:=(1/Pi)*Int(f(x)*cos(k*x),x=-Pi..Pi); # k=0,1,2,3,...
```

$$a_k := \frac{1}{\pi} \int_{-\pi}^{\pi} f(x) \cos(k\,x)\,dx$$

```
> a[0]:=simplify(subs(k=0,%));
```

$$a_0 := \frac{1}{\pi} \int_{-\pi}^{\pi} f(x)\,dx$$

```
> b[k]:=(1/Pi)*Int(f(x)*sin(k*x),x=-Pi..Pi); # k=1,2,3,...
```

$$b_k := \frac{1}{\pi} \int_{-\pi}^{\pi} f(x) \sin(k\,x)\,dx$$

ODD and Even Functions

```
> odd_function:=f(x)=x;
```

$$odd\_function := f(x) = x$$

```
> A[0]:=value(subs(f(x)=x,a[0]));
```

$$A_0 := 0$$

```
> A[k]:=value(subs(f(x)=x,a[k]));
```

$$A_k := 0$$

For odd functions the coefficients A[k], k = 0,1,2.. are identical to zero.

```
> B[k]:=value(subs(f(x)=x,b[k]));
```

$$B_k := -\frac{2\,(-\sin(\pi\,k) + \cos(\pi\,k)\,k\,\pi)}{\pi\,k^2}$$

```
> B[k]:=subs({sin(Pi*k)=0,cos(Pi*k)=(-1)^k},%);
```

$$B_k := -\frac{2\,(-1)^k}{k}$$

```
> FOURIER_series[f(x)=x][k=4]:=sum(B[k]*sin(k*x),k=1..4);
```

$$FOURIER\_series_{f(x)=x_{k=4}} := 2\sin(x) - \sin(2\,x) + \frac{2}{3}\sin(3\,x) - \frac{1}{2}\sin(4\,x)$$

```
> for i in [2,4,5] do FOURIER_series[f(x)=x][k=i]:=
  subs(k=i,sum(B[k]*sin(k*x),k=1..i)) od;
```

$$FOURIER\_series_{f(x)=x_{k=2}} := 2\sin(x) - \sin(2\,x)$$

$$FOURIER\_series_{f(x)=x_{k=4}} := 2\sin(x) - \sin(2\,x) + \frac{2}{3}\sin(3\,x) - \frac{1}{2}\sin(4\,x)$$

$$FOURIER\_series_{f(x)=x_{k=5}} := 2\sin(x) - \sin(2\,x) + \frac{2}{3}\sin(3\,x) - \frac{1}{2}\sin(4\,x) + \frac{2}{5}\sin(5\,x)$$

compact form:

```
> y(x,n)[f(x)=x]:=Sum(B[k]*sin(k*x),k=1..n);
```

$$y(x, n)_{f(x)=x} := \sum_{k=1}^{n} \left(-\frac{2\,(-1)^k \sin(k\,x)}{k} \right)$$

```
> y(x,4)[f(x)=x]:=value(subs(n=4,%));
```

$$y(x, 4)_{f(x)=x} := 2\sin(x) - \sin(2\,x) + \frac{2}{3}\sin(3\,x) - \frac{1}{2}\sin(4\,x)$$

```
> for i in [1,10,100] do
  y(x,n=i):=2*subs(n=i,sum((-1)^(n-1)*sin(n*x)/n,n=1..i)) od:
> plot({y(x,n=1),y(x,n=10),y(x,n=100)},x=-4*Pi..4*Pi,color=black,
  title="FOURIER-series with  n = [1, 10, 100]  for f(x) = x");
```




```
> even_function:=f(x)=x^2;
```

$$even\_function := \mathrm{f}(x) = x^2$$

```
> A[0]:=value(subs(f(x)=x^2,a[0]));
```

$$A_0 := \frac{2\,\pi^2}{3}$$

```
> A[k]:=value(subs(f(x)=x^2,a[k]));
```

$$A_k := \frac{2\,(-2\sin(\pi\,k) + k^2\sin(\pi\,k)\,\pi^2 + 2\cos(\pi\,k)\,k\,\pi)}{\pi\,k^3}$$

```
> A[k]:=subs({sin(Pi*k)=0,cos(Pi*k)=(-1)^k},%);
```

$$A_k := \frac{4\,(-1)^k}{k^2}$$

```
> B[k]:=value(subs(f(x)=x^2,b[k]));
```

$$B_k := 0$$

For even functions the coefficients B[k] are identical to zero.

```
> y(x,n)[f(x)=x^2]:=A[0]/2+Sum(A[k]*cos(k*x),k=1..n);
```

$$\mathrm{y}(x, n)_{\mathrm{f}(x) = x^2} := \frac{\pi^2}{3} + \left(\sum_{k=1}^{n} \left(\frac{4\,(-1)^k\cos(k\,x)}{k^2}\right)\right)$$

```
> y(x,4)[f(x)=x^2]:=value(subs(n=4,%));
```

$$y(x,4)_{f(x)=x^2} := \frac{\pi^2}{3} - 4\cos(x) + \cos(2x) - \frac{4}{9}\cos(3x) + \frac{1}{4}\cos(4x)$$

```
> for i in [1,10,100] do
  y(x,n=i):=Pi^2/3+subs(n=i,4*sum((-1)^n*cos(n*x)/n^2,
  n=1..i)) od:
> plot({y(x,n=1),y(x,n=10),y(x,n=100)},
  x=-4*Pi..4*Pi,color=black,
  title="FOURIER-series # n = [1, 10, 100] for f(x) = x^2");
```



Function f(x) = (x - X)^2 in several ranges:

```
> f(x)[-2*Pi]:=(x+2*Pi)^2;   x = [-3*Pi,  -Pi];
```

$$f(x)_{-2\pi} := (x + 2\pi)^2$$

$$x = [-3\pi, -\pi]$$

```
> f(x)[0]:=x^2;    x = [-Pi,  Pi];
```

$$f(x)_0 := x^2$$

$$x = [-\pi, \pi]$$

```
> f(x)[2*Pi]:=(x-2*Pi)^2;    x = [Pi,  3*Pi];
```

$$f(x)_{2\pi} := (x - 2\pi)^2$$

$$x = [\pi, 3\pi]$$

```
> alias(H=Heaviside,th=thickness,co=color):
> p[1]:=plot(f(x)[-2*Pi],x=-3*Pi..-Pi,0..Pi^2,th=2,co=black):
> p[2]:=plot(f(x)[0],x=-Pi..Pi,th=3,co=black):
> p[3]:=plot(f(x)[2*Pi],x=Pi..3*Pi,th=2,co=black):
> p[4]:=plot({Pi^2,Pi^2*H(x+3*Pi),Pi^2*H(x-3*Pi),
  Pi^2*H(x+2*Pi),Pi^2*H(x-2*Pi)},
  x=-3.001*Pi..3.001*Pi,co=black,
  title="Quadratic functions with the period  2*Pi"):
> plots[display]({seq(p[k],k=1..4)});
```




```
> constant_load:=q;
```

$$constant\_load := q$$

```
> A[0]:=value(subs(f(x)=q,a[0]));
```

$$A_0 := 2\,q$$

```
> A[k]:=value(subs(f(x)=q,a[k]));
```

$$A_k := \frac{2\sin(\pi\,k)\,q}{\pi\,k}$$

```
> limit_value:=Limit(A[k],k=0)=limit(%,k=0);
```

$$limit\_value := \lim_{k\to 0} \frac{2\sin(\pi\,k)\,q}{\pi\,k} = 2\,q$$

```
> B[k]:=value(subs(f(x)=q,b[k]));
```

$$B_k := 0$$

```
> y(x,n)[f=q]:=q+Sum((2*q*sin(Pi*k)/Pi/k)*cos(k*x),k=1..n);
```

$$\mathrm{y}(x,n)_{f=q} := q + \left(\sum_{k=1}^{n} \left(\frac{2\,q\sin(\pi\,k)\cos(k\,x)}{\pi\,k}\right)\right)$$

```
> y(x,n)[f=q]:=
  simplify(subs({sin(Pi*k)=0,cos(Pi*k)=(-1)^k},%));
```

$$\mathrm{y}(x,n)_{f=q} := q$$

The solution is trivial. For all k yields: y(x, k...n) = q.

```
> another_example:=f=1+x;
```

$$another\_example := f = 1 + x$$

```
> A[0]:=value(subs(f(x)=1+x,a[0]));
```

$$A_0 := 2$$

```
> A[k]:=value(subs(f(x)=1+x,a[k]));
```

$$A_k := \frac{2\sin(\pi\,k)}{\pi\,k}$$

```
> limit_value:=Limit(A[k],k=0)=limit(2*sin(Pi*k)/Pi/k,k=0);
```

$$limit\_value := \lim_{k \to 0} \frac{2\sin(\pi k)}{\pi k} = 2$$

```
> B[k]:=value(subs(f(x)=1+x,b[k]));
```

$$B_k := -\frac{2\,(-\sin(\pi k) + \cos(\pi k)\,k\,\pi)}{\pi k^2}$$

```
> B[k]:=subs({sin(Pi*k)=0,cos(Pi*k)=(-1)^k},%);
```

$$B_k := -\frac{2\,(-1)^k}{k}$$

```
> y(x,n)[f=1+x]:=1+Sum(B[k]*sin(k*x),k=1..n);
```

$$y(x, n)_{f=1+x} := 1 + \left(\sum_{k=1}^{n} \left(-\frac{2\,(-1)^k \sin(k\,x)}{k}\right)\right)$$

```
> y(x,4)[f=1+x]:=value(subs(n=4,%));
```

$$y(x, 4)_{f=1+x} := 1 + 2\sin(x) - \sin(2\,x) + \frac{2}{3}\sin(3\,x) - \frac{1}{2}\sin(4\,x)$$

```
> for i in [1,4,100] do
  y(x,n=i):=1+subs(n=i,sum(2*sin(Pi*n)*cos(n*x)/Pi/n-
  (2*(-sin(Pi*n)+cos(Pi*n)*Pi*n)*sin(n*x))/Pi/n^2,
  n=1..i)) od:
> plot({1,y(x,n=1),y(x,n=4),y(x,n=100)},x=-4*Pi..4*Pi,co=black,
  title="FOURIER series # n = [1, 4, 100] for f(x) = 1+x");
```

FOURIER series # n = [1, 4, 100] for f(x) = 1+x



***FOURIER* Representation of the *HEAVISIDE* Function**

```
> alias(H=Heaviside):
> F(x)[H_odd]:=H(x)+2*sum((-1)^n*H(x-n*Pi),n=1..N)-
  H(-x)+2*sum((-1)^n*H(x+n*Pi),n=1..N);
```

$$F(x)_{H\_odd} := H(x) + 2\left(\sum_{n=1}^{N} (-1)^n H(x - \pi n)\right) - H(-x) + 2\left(\sum_{n=1}^{N} (-1)^n H(x + \pi n)\right)$$

```
> f(x)[H_odd]:=subs(N=4,%);
```

$$f(x)_{H\_odd} := H(x) + 2\left(\sum_{n=1}^{4} (-1)^n H(x - \pi n)\right) - H(-x) + 2\left(\sum_{n=1}^{4} (-1)^n H(x + \pi n)\right)$$

```
> plot(%,x=-4.5*Pi..4.5*Pi,co=black);
```




```
> A[0]:=value(subs(f(x)=f(x)[H_odd],a[0]));
```

$$A_0 := 0$$

```
> A[k]:=value(simplify(subs(f(x)=f(x)[H_odd],a[k])));
```

$$A_k := 0$$

```
> B[k]:=simplify(value(subs(f(x)=f(x)[H_odd],b[k])));
```

$$B_k := -\frac{2(\cos(\pi k) - 1)}{\pi k}$$

```
> B[k]:=subs(cos(Pi*k)=(-1)^k,%);
```

$$B_k := -\frac{2((-1)^k - 1)}{\pi k}$$

```
> y(x,n)[f=H_odd]:=sum(B[k]*sin(k*x),k=1..n);
```

$$y(x, n)_{f=H\_odd} := \sum_{k=1}^{n} \left(-\frac{2((-1)^k - 1)\sin(k x)}{\pi k}\right)$$

```
> y(x,4)[H_odd]:=value(subs(n=4,%));
```

$$y(x, 4)_{H\_odd} := \sum_{k=1}^{4} \left(-\frac{2((-1)^k - 1)\sin(k x)}{\pi k}\right)$$

```
> for i in [1,2,3,4,5,6] do
  y(x,n=i):=subs(n=i,2*sum(((1-(-1)^n)/Pi/n)*sin(n*x),
  n=1..i)) od;
```

$$y(x, n = 1) := \frac{4\sin(x)}{\pi}$$

$$y(x, n = 2) := \frac{4\sin(x)}{\pi}$$

$$y(x, n = 3) := \frac{4 \sin(x)}{\pi} + \frac{4}{3} \frac{\sin(3 x)}{\pi}$$

$$y(x, n = 4) := \frac{4 \sin(x)}{\pi} + \frac{4}{3} \frac{\sin(3 x)}{\pi}$$

$$y(x, n = 5) := \frac{4 \sin(x)}{\pi} + \frac{4}{3} \frac{\sin(3 x)}{\pi} + \frac{4}{5} \frac{\sin(5 x)}{\pi}$$

$$y(x, n = 6) := \frac{4 \sin(x)}{\pi} + \frac{4}{3} \frac{\sin(3 x)}{\pi} + \frac{4}{5} \frac{\sin(5 x)}{\pi}$$

```
> for i in [1,5,99] do
  y(x,n=i):=subs(n=i,2*sum(((1-(-1)^n)/Pi/n)*sin(n*x),
  n=1..i)) od:
> plot({f(x)[H_odd],y(x,n=1),y(x,n=5),y(x,n=99)},
  x=-4.5*Pi..4.5*Pi,co=black);
```



Interval (0, 2*Pi):

```
> y(x):=alpha[0]/2+
  Sum(alpha[k]*cos(k*x)+beta[k]*sin(k*x),k=1..infinity);
```

$$y(x) := \frac{1}{2} \alpha_0 + \left(\sum_{k=1}^{\infty} (\alpha_k \cos(k x) + \beta_k \sin(k x)) \right)$$

```
> alpha[k]:=(1/Pi)*Int(phi(x)*cos(k*x),x=0..2*Pi);
  # k = 0,1,2,3,...
```

$$\alpha_k := \frac{1}{\pi} \int_0^{2\pi} \phi(x) \cos(k x) \, dx$$

```
> alpha[0]:=simplify(subs(k=0,%));
```

$$\alpha_0 := \frac{1}{\pi} \int_0^{2\pi} \phi(x) \, dx$$

```
> beta[k]:=(1/Pi)*Int(phi(x)*sin(k*x),x=0..2*Pi);
  # k = 1,2,3,...
```

$$\beta_k := \frac{1}{\pi}\int_0^{2\pi} \phi(x)\sin(k\,x)\,dx$$

```
> alias(H=Heaviside,th=thickness,co=color):
> phi(x,N):=H(x)+2*Sum((-1)^n*H(x-n*Pi),n=1..N);
```

$$\phi(x, N) := \mathrm{H}(x) + 2\left(\sum_{n=1}^{N} (-1)^n\,\mathrm{H}(x - \pi\,n)\right)$$

```
> phi(x,4):=subs(N=4,%);
```

$$\phi(x, 4) := \mathrm{H}(x) + 2\left(\sum_{n=1}^{4} (-1)^n\,\mathrm{H}(x - \pi\,n)\right)$$

```
> plot(%,x=0..4.5*Pi,co=black);
```




```
> Alpha[0]:=value(subs({phi(x)=phi(x,4),k=0},
  alpha[k]));
```

$$\mathrm{A}_0 := 0$$

```
> Alpha[k]:=simplify(value(subs(phi(x)=phi(x,4),
  alpha[k])));
```

$$\mathrm{A}_k := -\frac{2\sin(\pi\,k)\,(\cos(\pi\,k) - 1)}{\pi\,k}$$

```
> A[k]:=subs({sin(Pi*k)=0,cos(Pi*k)=(-1)^k},%);
```

$$A_k := 0$$

```
> BETA[k]:=simplify(value(subs(phi(x)=phi(x,4),
  beta[k])));
```

$$BETA_k := \frac{2\cos(\pi\,k)\,(\cos(\pi\,k) - 1)}{\pi\,k}$$

```
> BETA[k]:=subs(cos(Pi*k)=(-1)^k,%);
```

$$BETA_k := \frac{2\,(-1)^k\,((-1)^k - 1)}{\pi\,k}$$

```
> y(x,n):=Sum(BETA[k]*sin(k*x),k=1..n);
```

$$\mathrm{y}(x, n) := \sum_{k=1}^{n}\left(\frac{2\,(-1)^k\,((-1)^k - 1)\sin(k\,x)}{\pi\,k}\right)$$

```
> y(x,4):=value(subs(n=4,%));
```

$$y(x,4) := \frac{4\sin(x)}{\pi} + \frac{4}{3}\frac{\sin(3x)}{\pi}$$

```
> for i in [1,3,99] do
  y(x,n=i):=subs(n=i,y(x,n)) od:
> plot({y(x,n=1),y(x,n=3),y(x,n=99)},
  x=-4*Pi..8*Pi,co=black);
```



Continuous Functions with Cusps

```
> h(x):=piecewise(x>=0 and x<=1,x,x>=1 and x<=2*Pi,1);
```

$$h(x) := \begin{cases} x & 0 \le x \textbf{ and } x \le 1 \\ 1 & 1 \le x \textbf{ and } x \le 2\pi \end{cases}$$

```
> plot(h(x),x=0..2*Pi,co=black);
```




```
> Alpha[k]:=simplify(value(subs(phi(x)=h(x),alpha[k])));
```

$$A_k := \frac{-1 + \cos(k) + 2\,k\sin(\pi k)\cos(\pi k)}{k^2\,\pi}$$

```
> Alpha[k]:=subs(sin(k*Pi)=0,%);
```

$$A_k := \frac{-1 + \cos(k)}{k^2\,\pi}$$

```
> Alpha[0]:=simplify(value(subs(phi(x)=h(x),alpha[0])));
```

$$A_0 := \frac{-1 + 4\pi}{2\pi}$$

```
> BETA[k]:=simplify(value(subs(phi(x)=h(x),beta[k])));
```

$$BETA_k := -\frac{-\sin(k) + 2\,k\cos(\pi\,k)^2 - k}{k^2\,\pi}$$

```
> BETA[k]:=subs((cos(k*Pi))^2=1,%);
```

$$BETA_k := -\frac{-\sin(k) + k}{k^2\,\pi}$$

```
> y(x,n):=Alpha[0]/2+sum(Alpha[k]*cos(k*x)+
  BETA[k]*sin(k*x),k=1..n);
```

$$y(x, n) := \frac{-1 + 4\,\pi}{4\,\pi} + \left(\sum_{k=1}^{n} \left(\frac{(-1 + \cos(k))\cos(k\,x)}{k^2\,\pi} - \frac{(-\sin(k) + k)\sin(k\,x)}{k^2\,\pi}\right)\right)$$

```
> y(x,1):=evalf(subs(n=1,%),4);
```

$$y(x, 1) := 0.9205 - 0.1463\cos(x) - 0.05046\sin(x)$$

```
> y(x,3):=evalf(subs(n=3,%%),4);
```

$$y(x, 3) := 0.9205 - 0.1463\cos(x) - 0.05046\sin(x) - 0.1127\cos(2.\,x) - 0.08680\sin(2.\,x) - 0.07038\cos(3.\,x) - 0.1011\sin(3.\,x)$$

```
> y(x,99):=evalf(subs(n=99,%%%),4):
> for i in [1,3,99] do
  y(x,n=i):=subs(n=i,y(x,n)) od:
> plot({y(x,n=1),y(x,n=3),y(x,n=99)},
  x=0..2*Pi,color=black);
```



L-two Norm

```
> L_two[n]:=sqrt((1/2/Pi)*Int((H(x)-Y(x,n))^2,
  x=0..2*Pi));
```

$$L\_two_n := \frac{1}{2}\sqrt{2}\,\sqrt{\frac{1}{\pi}\int_0^{2\,\pi} (\mathrm{H}(x) - \mathrm{Y}(x, n))^2\,dx}$$

```
> for i in [1,3,99] do
  L_two[n=i]:=evalf(sqrt((1/2/Pi)*value(int((h(x)-y(x,i))^2,
  x=0..2*Pi))),4) od;
```

$$L\_two_{n=1} := 0.1865$$

$$L\_two_{n=3} := 0.1305$$

$$L\_two_{n=99} := 0.02257$$

For n = 99 the *FOURIER* series y(x, n = 99) represents a good approximation to the above given function h(x).

```
> g(x):=piecewise(x<-1,-1,x>-1 and x<1,x,x>1,1);
```

$$g(x) := \begin{cases} -1 & x < -1 \\ x & -1 < x \textbf{ and } x < 1 \\ 1 & 1 < x \end{cases}$$

```
> plot(g(x),x=-Pi..Pi,co=black);
```




```
> A[k]:=simplify(value(subs(f(x)=g(x),a[k])));
```

$$A_k := 0$$

```
> A[0]:=value(subs(f(x)=g(x),a[0]));
```

$$A_0 := 0$$

```
> B[k]:=simplify(value(subs(f(x)=g(x),b[k])));
```

$$B_k := -\frac{2\,(\cos(\pi\,k)\,k - \sin(k))}{k^2\,\pi}$$

```
> B[k]:=subs(cos(Pi*k)=(-1)^k,%);
```

$$B_k := -\frac{2\,((-1)^k\,k - \sin(k))}{k^2\,\pi}$$

```
> y(x,n):=sum(B[k]*sin(k*x),k=1..n);
```

$$y(x, n) := \sum_{k=1}^{n} \left(-\frac{2\,((-1)^k\,k - \sin(k))\,\sin(k\,x)}{k^2\,\pi} \right)$$

```
> y(x,1):=evalf(value(subs(n=1,%)),4);
```

$$y(x, 1) := 1.172\,\sin(x)$$

```
> y(x,3):=evalf(value(subs(n=3,%%)),4);
```

$$y(x, 3) := 1.172\,\sin(x) - 0.1736\,\sin(2.\,x) + 0.2222\,\sin(3.\,x)$$

```
> for i in[1,3,99] do y(x,n=i):=subs(n=i,y(x,n)) od:
> plot({y(x,n=1),y(x,n=3),y(x,n=99)},
  x=-2*Pi..2*Pi,co=black);
```




```
> L_two[n]:=sqrt((1/Pi)*Int((G(x)-Y(x,n))^2,x=0..Pi));
```

$$L\_two_n := \sqrt{\frac{1}{\pi}\int_0^{\pi}(G(x)-Y(x,n))^2\,dx}$$

```
> for i in [1,3,99] do
  L_two[n=i]:=evalf(sqrt((1/Pi)*int((g(x)-y(x,i))^2,
  x=0..Pi)),4) od;
```

$$L\_two_{n=1} := 0.3172$$

$$L\_two_{n=3} := 0.2468$$

$$L\_two_{n=99} := 0.02264$$

>

For n = 99 the *FOURIER* series y(x, n = 99) represents a good approximation to the above given function g(x).

>